Getting Started with HSPICE

Topics

- Netlist file structure
- HSPICE output files
- Using components, sources and subcircuits to construct a netlist
- HSPICE analysis types
- Simulation controls and options

Netlist Structure

Title	`First line is always the title`
Comment character	`* - comment for a line` `$ - comment after a command`
Options	`.option post`

Print/Probe/Analysis	`.print v(d) i(rl)`
	`.probe v(g)`
	`.tran .1n 5n`

Initial Conditions	`.ic v(b) = 0 $ input state`
Sources	`Vg g 0 pulse 0 1 0 0.15 0.15 0.42` `* example of a voltage source`
Circuit Description	`MN d g gnd n nmos` `RL vdd d 1K`

Models	`.model n nmos level = 49`
	`+ vto = 1 tox = 7n`
	`* '+' continuation character`

End	`.end $ terminates the simulation`

Netlist structure

- A netlist consists of one main file and one or more optional added files
- Some commands can be used to structure the netlist:
  - Add a file to your netlist

    ```
    .INCLUDE 'filename'
    ```

  - Add a library to the netlist

    ```
    .LIB 'filename'libname
    ```

  - Assign a parameter value

    ```
    .PARAM  res=1
    .PARAM  res2='res+1'
    R1 n1 n2 'res2'
    ```

  - Alter the netlist or parameters and run more simulations

    ```
    .ALTER
    .DATA
    ```

- Elements in a netlist are order independent
  - For Parameters and Options the last definition is used

Netlist Topologies to avoid



No voltage loops



No dangling nodes



Every node must have a DC path to ground



No stacked current sources



No ideal voltage source in closed inductor loop



No ideal current source in closed capacitor loop

Files and suffixes

HSPICE Input	
Input netlist	.sp
Model libraries	.inc, .lib
HSPICE Output	
Run Status	.st0
Output listing	.lis
Analysis data, transient	.tr# (e.g. .tr0)
Analysis data, dc	.sw# (e.g. .sw0)
Analysis data, ac	.ac# (e.g. .ac0)
FFT Analysis data	.ft# (e.g. .ft0)
Measure output	.m*# (e.g. .mt0)
Waveform Viewer Input	
All analysis data and measure output	

The output list file

- .lis contains results of
  - .PRINT
  - .OP
  - .OPTIONS

Units and Scale factors

- Element units:
  - R - ohm
  - C - Farad
  - L - Henry
- Technology Scaling:
  - SCALE and SCALM
  - ALL lengths and widths are in METERS
- Common conversion factors

  MIL(S) = 25.4e-6

  FT = .3048 (METERS)

  DB = 20log10

- Scale Factors

  A = 1e-18

  F = 1e-15

  P = 1e-12

  N = 1e-9

  U = 1e-6

  M = 1e-3

  K = 1e3

  MEG = X = 1e6

  G = 1e9

  T = 1e12

Circuit Elements

- Passive components
    - R, C, L, etc.
- Sources
    - Independent sources
        - DC, AC, TRAN (pulse, pwl, pattern, pseudo random bit stream, sin, AM, exp, single frequency frequency modulated), Mixed, Digital
    - Dependent sources
        - G (voltage controlled current source), E (voltage controlled voltage source), F (current controlled voltage source), H (current controlled current source)
- Active Elements
    - D, M, Q, J (need a .MODEL)

Subcircuits

- Subcircuit definition syntax:

  `.SUBCKT subnam n1 n2 [n3 n4 … ] [parnam=val … ]`

  – `subnam` – Reference name for the subcircuit model call

  – `n1, n2 …` - Node numbers for external reference

    - Any element nodes appearing in the subcircuit, but not included in this list, are strictly local, EXCEPT
      – Ground node (0)
      – Nodes assigned using BULK (MOSFET) or SUBSTRATE (BJT)
      – Nodes assigned using the `.GLOBAL` statement

  – `parnam=val` - A parameter name set to a value

    - For use only in the subcircuit
    - Overridden by an assignment in the subcircuit call or by a value set in a .PARAM statement that is external to the subcircuit

Subcircuit Example



Analysis Types

- DC Operating Point
- DC Analysis
- AC Analysis
- Transient Analysis
- Temperature Analysis

Output Commands

- `.PROBE`:
  - Save output variables in analysis data file without additional output in the output listing file
  - With option probe, can limit *.tr# file size
- `.PRINT`:
  - Numeric analysis results printed to output listing file
- `.MEASURE`:
  - Print numeric results of measured electrical specifications for specified analysis

Examples:

```
.PROBE tran v(4) i(vin) par('v(out)/v(in)')
.PRINT AC VM(4,2) VP(8,3) VR(7) Ii(R1)
.PRINT LX8(m1)
```

- Print the drain-source conductance of element m1

Limiting the size of the output data

- Specifying analysis data format:
  - `.OPTION POST` → Generate binary output waveform files
- Limiting the size of the analysis data file:
  - `.OPTION PROBE`
    - ALL nodes plotted by default
    - Limit data in analysis data file to that specified in `.PRINT` and `.PROBE` statements
  - `.OPTION INTERP`
    - Limit the number of points stored to step size specified in `.TRAN` statement
    - Pre-interpolates the output to the interval specified on the `.TRAN` statement

Output Variables

- Five Groups of Output Variables:
  - DC and transient analysis
    - Displays individual nodal voltages, branch currents, element power dissipation
  - AC analysis
    - Displays imaginary and real components of nodal voltage, branch current
    - Displays phase and impedance parameters
  - Element templates
    - Displays element specific nodal voltages, branch currents, element parameters, and the derivatives of element voltage, current, or charge
  - .MEASURE
    - Displays user-defined variables as specified in the .MEASURE statement
  - Parametric statements
    - Displays mathematically, user-defined expressions operating on nodal voltages, etc

Output Variables: Examples

- DC and Transient output examples:
  - Standard form is V(node) or I(element)

    `v(1)`
    - Voltage at node 1

    `i(Rin)`
    - Current through Rin (direction of I is n1 to n2)

    `v(1,2)`
    - Voltage between node 1 and node 2 (differential)
  - Complex addressing
    - `i1(xinv1.m3)`
    - Drain current of transistor m3 in subcircuit inv1

Output Variables: Examples

- DC and Transient output examples (cont.)

  `p(rload)`
  – Power dissipated in rload at point of analysis

  `p(m1)`
  – Power dissipated in transistor m1 at point of analysis

  `Power`
  – Total power dissipation output at point of analysis

  `v(x3.5)`
  – Voltage at INTERNAL node 5 of subckt x3

  `par('p(x1.m1)+p(x2.m2)')`
  – Sum of power in m1 of x1 and m2 of x2

  `i3(2:q2)`
  – Emitter current of q2 in second subcircuit called

Output Variables: Examples

- AC analysis output examples:

  `vi(2)`

  – Imaginary voltage component at node 2

  `ip1(q4)`

  – The phase of the collector current in q4

  `vdb(2,8)`

  – The voltage ratio between node 2 and 8 in decibels

  `vp(4,6)`

  – The arctangent [vi(4,6)/vr(4,6)]

  `vp(6)`

  – Phase at node 6

ACOUT option (1/2)

- ACOUT controls the AC output calculation method, for the difference in values of magnitude, phase, and decibels:
  - `ACOUT = 1` (default) selects HSPICE method, which calculates the difference of the magnitudes of the values
  - Real and imaginary

    ```
    VR(N1,N2) = REAL [V(N1,0)] - REAL [V(N2,0)]
    VI(N1,N2) = IMAG [V(N1,0)] - IMAG [V(N2,0)]
    ```

  - Magnitude

    $$VM(N1,0) = [VR(N1,0)^2 + VI(N1,0)^2]^{0.5}$$
    $$VM(N2,0) = [VR(N2,0)^2 + VI(N2,0)^2]^{0.5}$$

    ```
    VM(N1,N2) = VM(N1,0) - VM(N2,0)
    ```

  - Phase

    ```
    VP(N1,0) = ARCTAN[VI(N1,0)/VR(N1,0)]
    VP(N2,0) = ARCTAN[VI(N2,0)/VR(N2,0)]
    VP(N1,N2) = VP(N1,0) - VP(N2,0)
    ```

  - Decibel

    ```
    VDB(N1,N2) = 20·LOG10(VM(N1,0)/VM(N2,0))
    ```

ACOUT option (2/2)

Berkeley SPICE method

- ACOUT = 0 selects SPICE method, which calculates the magnitude of differences
- Real and imaginary

  VR(N1,N2) = REAL[V(N1,0) - V(N2,0)]

  VI(N1,N2) = IMAG[V(N1,0) - V(N2,0)]
- Magnitude

  $VM(N1,N2) = [VR(N1,N2)^2 + VI(N1,N2)^2]^{0.5}$
- Phase

  VP(N1,N2) = ARCTAN[VI(N1,N2)/VR(N1,N2)]
- Decibel

  $VDB(N1,N2) = 20 \cdot LOG10[VM(N1,N2)]$

ACOUT Example

```
.option ACOUT=1
v1 1 0 ac .5
v2 0 2 ac .5
r1 1 0 1k
r2 2 0 1k
.print ac Vm(1) vm(2) vm(1,2) vp(1,2)
```

- output:
  0.5 0.5 0 180
- If we change to .option ACOUT=0 then

  0.5 0.5 1 0

Output Variables: Element Templates

- Element Templates:
  - Display element specific nodal voltages, branch currents, element parameters and the derivatives of element voltage, current, and charge

Examples:

```
.print tran lv16(m3)
```

- Prints the effective drain conductance (1/rdeff)

There are over hundred element templates: check documentation

Output variables: .MEASURE

- .MEASURE
  - Prints user-defined electrical specifications of a circuit
  - Used extensively in optimization
  - Has seven fundamental measurement modes, each with its own form:
    - Rise, fall, and delay
    - Average, RMS, min, max, and P-P
    - Find - When
    - Equation evaluation
    - Derivative evaluation
    - Integral evaluation
    - Relative error

.MEASURE Examples

- Measure delay from 2nd rising edge of node 1 to 2nd falling edge of node 2:

```
.MEAS TRAN TDELAY TRIG V(1) VAL=2.5 TD=10ns RISE=2
+                 TARG V(2) VAL=2.5         FALL=2
```

- Measure the average voltage of node 10 between 10ns and 55ns:

```
.MEAS TRAN avgval AVG V(10) From=10ns To=55ns
```

- Measure the time of the 5th rise of node "osc_out" at 2.5v and report as "r5":

```
.MEAS TRAN r5 WHEN V(osc_out)=2.5v RISE=5
```

- Measure v(out) when v(in)=40m, then store and print variable "result":

```
.MEAS TRAN result FIND v(out) WHEN v(in)=40m
```

- Make a calculation on previously measured values:

```
.MEAS TRAN my_result PARAM='(result/avgval)'
```

Output variables: Parametric Output

Parametric Variables are waveform vectors

- Parametric Output Variables:

```
.PRINT tran PAR('algebraic expression')
.PRINT tran out_var=PAR('algebraic expression')
```

Examples:

```
.PRINT dc conductance=PAR('i(m1)/v(22)')
.PROBE tran PAR('log10(i(xff4.m1))')
.PRINT tran vds=PAR(vds) vgs=PAR(vgs)
```

Commonly used .OPTIONS

See the Command Reference for a complete list of options

Listing
ACCT
BRIEF
INGOLD
LIST
NODE
NOMOD
NUMDGT
OPTLST

Interface
ARTIST
PSF
CSDF
POST
PROBE

Output
CAPTAB
DCCAP
UNWRAP

Error
BADCHR
DIAGNOSTIC

Convergence
CONVERGE
DV
GMIN
GMINDC
GRAMP

Model
SCALE
SEARCH
TNOM

MOSFETs
SCALM
WL

DC Matrix
ITL1

Transient Accuracy
RUNLVL
ACCURATE

Efficiency
AUTOSTOP
NOELCK
NOTOP

Time Step
DELMAX
METHOD

General Listing Options

`.option LIST`

- Prints a list of netlist elements, node connections and, parameter values
- Prints effective sizes of elements and key values
- Useful in diagnosing topology-related problems

`.option NODE`

- Prints a node connection table
- The nodal cross-reference table lists each node and all the elements connected to it
- Useful in diagnosing topology related non-convergence problems

`.option ACCT`

- Reports job accounting and runtime statistics at the end of the output listing
- Useful in observing simulation efficiency

`.option OPTS`

- Reports settings of all `.OPTIONS` in the listing file

More listing file options

- Simulation listing can get large:
- . OPTION nomod
  suppress the printout of model parameters
- .OPTION BRIEF
  suppress the printout of the netlist
- But … if you have troubles, remove the nomod and brief options so you can see all information

More listing file options

`.option INGOLD=value`

- Controls the format of numbers in printouts
  - 0 – engineering format (default)
  - 1 – mix of fixed and exponential format
  - 2 – exponential format

`.option NUMDGT=value`

- Number of significant digits printed for output variables
  - Range is between 1 to 10
  - Defaults is 4

Output Control Options

- POST (syntax: .OPTION POST)
- PROBE (syntax: .OPTION PROBE)

Save simulation results for a waveform viewer

Limit the waveforms generated

- POSTTOP
  - Limits data written to waveform file to top-level nodes only
- Syntax

  `.OPTION POSTTOP`
- POSTLVL
  - Limits data written to waveform file that only a specified level of nodes signals will be saved
- Syntax

  `.OPTION POSTLVL=n`
  - `n=1` top level only
  - `n=0` all levels, i.e., POSTLVL option is turned off

Netlist Options (1/2)

- GLOBAL SCALE
  - Scale element geometric instance parameters whose default unit is meters.
- SYNTAX

```
.option scale=value
```

Example:

```
.option scale=1e-6
M1 Vdd 10 20 0 mymodel L=1u w=1u    $ L=1e-12, W=1e-12
M2 Vdd 10 20 0 mymodel L=1 w=1      $ L=1e-6, W=1e-6
C1 Vdd 10 capmod L=0.1 W=0.1        $ L=1e-7, W=1e-7
R1 20 0 resmod L=1 W=1              $ L=1e-6, W=1e-6
```

Netlist Options (2/2)

- LOCAL SCALE
  - Scale passive device values

Examples:

```
R1 1 0 1k scale=10    $ R1=10K
C1 5 0 1u scale=10    $ C1=10u
L1 10 0 1u scale=10  $ L1=10u
```

Model Options (1/2)

M = Models

- GLOBAL SCALM
  - Affects model parameters with units in METERS
  - Scales active devices (MOSFETs, diodes, JFETs etc.)
  - Passive devices are not affected by scalm
  - Local scalm takes precedence over global

```
.option scalm
```

- SYNTAX:

```
.option scalm=value
```

Example:

```
.option scalm=1e-6
```

Model Options (2/2)

- LOCAL SCALM
  - Affects model parameters with the units of meters

Example:

```
M1 Vdd 10 20 0 mymodel L=1u w=1u
.model mymodel nmos scalm=1e-6 level=2
+ tox=1e-1 kp=2.0e-5
```

The units of Kp are A/V^2

  - Only model parameters tox will be scaled

```
tox=1e-7
```

  - L and W will not be affected